ADDITIONAL MOON TRACKING COMPUTER AND CALCULATOR PROGRAMS

DIVISION OF VARIAN
301 Industrial Way
San Carlos, California

AS-49-17

4156

MODIFICATIONS TO HP2000C PROGRAM

In bulletin AS-49-13, there is a moon tracking program written for the HP2000C computer. In that particular program the equations for elevation were based on the center of the earth. As a result, there is about a 1^{o} error in the elevation number. By changing one statement and adding another, the error can be reduced to under 0.2^{o}. The following statements show the program before and after the changes were made. Also, in statement numbers 280 and 350 the line reads FOR N=1 to 25. If these statement lines are changed to FOR N=1 to 31, then 31 days of data can be requested instead of being limited to 25 days.

BEFORE

```
1300  LET H=L6-G
1310  REM:CALCULATION OF ELEVATION,E,OF OBJECT
1320  LET E3=COS(L5)*COS(H)*COS(D1)+SIN(D1)*SIN(L5)
1330  LET E2=SQR(1-(E3*E3))
1340  LET E=ATN(E3/E2)
1350  IF E<0 THEN 1810
1360  IF E>I6*R5 THEN 1810
1370  REM:CALCULATION OF AZIMUTH,A,OF OBJECT
1380  LET A2=SIN(D1)/(COS(L5)*COS(E))
1390  LET A1=SIN(L5)*SIN(D1)+COS(L5)*COS(D1)*COS(H)
```

AFTER

```
1300  LET H=L6-G
1310  REM:CALCULATION OF ELEVATION,E,OF OBJECT
1320  LET E3=COS(L5)*COS(H)*COS(D1)+SIN(D1)*SIN(L5)
1330  LET E2=SQR(1-(E3*E3))
1340  LET E=ATN((E3/E2)-(1/(61.33*E2)))
1345  LET F=ATN(E3/E2)
1350  IF E<0 THEN 1810
1360  IF E>I6*R5 THEN 1810
1370  REM:CALCULATION OF AZIMUTH,A,OF OBJECT
1380  LET A2=SIN(D1)/(COS(L5)*COS(F))
1390  LET A2=A2-(SIN(L5)/COS(L5))*(SIN(F)/COS(F))
```

The following program is a version of Lance Collister's (WB7CCI) original moon tracking program written in GE BASIC. This particular version was modified by Jay Liebmann, K5JL, to run on a Mits Altair 8800B. The program requires about 6700 bytes of memory. For those with a smaller memory, Jay has a stripped down version of the program which requires only 4000 bytes.

```
500 DIM F(31),V(31),Y(31),Q(31),S(31)
560  P5=2.0000000000*3.1415926535
570  D5=360.0000000000/P5
580 R5=P5/360.0000000000
582 DEF FNA(X)=INT(X*D5*10+.5)/10
584 DEF FNB(X)=(X-INT(X))*P5
585 PRINT"WHAT ARE THE STATION CALL LETTERS";
587 INPUT W$
590 PRINT"WHAT IS YOUR LATITUDE IN DEGREES,MINUTES";
600 INPUT L5,U5
610 PRINT"WHAT IS YOUR LONGITUDE IN DEGREES,MINUTES";
620 INPUT L6,U6
630  L5=(L5+U5/60)*R5
640  L6=(L6+U6/60)*R5
650 PRINT"WHAT IS DESIRED PRINTING INCREMENT IN MINUTES";
660 INPUT I
670 PRINT "DO YOU ONLY WANT PRINTOUT WHEN THE MOON IS NEAR THE HORIZON";
690 INPUT B$
700 IF B$="YES" THEN 730
710 LET I6=100
720 GOTO 800
730 PRINT"BELOW WHAT ELEVATION IN DEGREES DO YOU WANT PRINTOUT"
731 PRINT"TO OCCUR";
740 INPUTI6
750 PRINT"WHAT ARE THE GMT MONTH, DAY, YEAR DESIRED";
760 FOR N=1 TO 31
770 INPUT F(N),V(N),Y(N)
780 IF F(N)=0 THEN 860
785 NEXT N
790 GOTO 760
800 PRINT"WHAT ARE THE GMT MONTH, DAY, YEAR, TIME BEGINNING, TIME ENDING
820 FOR N= 1 TO 31
830 INPUT F(N),V(N),Y(N),Q(N),S(N)
840 IF F(N)=0 THEN 860
845 NEXT N
850 GOTO 820
860  N5=N-1
870 FOR N=1 TO N5
880 IF B$="YES" THEN 900
890 GOTO 930
900  E1=2400
910  B=0
920 GOTO 950
930  E1=S(N)
940  B=Q(N)
950  M=F(N)
960  D=V(N)
970  Y=Y(N)
980  Y1=Y-(INT(Y/100)*100)
990 PRINT
```

```
1000 PRINT
1010 PRINT"POSITION OF THE MOON ON";M;"/";D;"/";Y1;"GMT  FROM""  "W$
1020 PRINT
1030 PRINT"  GMT","  AZ","  EL","  GHA","  DEC"
1040 PRINT"  ---","  --","  --","  ---","  ---"
1050 PRINT
1060 I1=2
1080 IFM>=3 THEN 1160
1090 IF INT((Y-1853)/4)<11 THEN 1120
1100  C1=-1
1110 GOTO 1130
1120  C1=0
1130  J1=365*(Y-1853)+D+30*(M+9)+INT((M+10)/2)
1140  J2=INT((Y-1853)/4)+1+C1
1150 GOTO 1270
1160 IF INT((Y-1852)/4)<11 THEN 1190
1170  C1=-1
1180 GOTO 1200
1190  C1=0
1200 IF M=9 THEN 1240
1210 IF M=11 THEN 1240
1220  C2=0
1230 GOTO 1250
1240  C2=1
1250  J1=365*(Y-1852)+D+30*(M-3)+INT((M-2)/2)
1260  J2=INT((Y-1852)/4)+C1+C2
1270  J=J1+J2
1280  T1=J-17472.5
1290  D9=(B-INT(B/100)*100)+INT(B/100)*60
1300 D6=(E1-INT(E1/100)*100)+INT(E1/100)*60
1310  D7=D9-D6
1320  D8=D7-I
1330 IF D7>0 THEN 1350
1340 GOTO 1380
1350 IF D8>=0 THEN 2220
1360  B=E1
1380 T=(B-INT(B/100)*100)/1440+INT(B/100)/24
1390  T5=T1+T
1400  K1=FNB(.751213+.036601102*T5)
1410  K2=FNB(.822513+.0362916457*T5)
1420  K3=FNB(.995766+.00273777852*T5)
1430  K4=FNB(.974271+.0338631922*T5)
1440  K5=FNB(.0312525+.0367481957*T5)
1450  L8=K1+.658*R5*SIN(2*K4)+6.289*R5*SIN(K2)
1460  L8=L8-1.274*R5*SIN(K2-2*K4)-.186*R5*SIN(K3)
1470  L8=L8+.214*R5*SIN(2*K2)-.114*R5*SIN(2*K5)
1480  L8=L8-.059*R5*SIN(2*K2-2*K4)-.057*R5*SIN(K2+K3-2*K4)
1490 K6=K5+.6593*R5*SIN(2*K4)+6.2303*R5*SIN(K2)-1.272*R5*SIN(K2-2*K4)
1500  L7=5.144*R5*SIN(K6)-.146*R5*SIN(K5-2*K4)
1520  LETD1=COS(L7)*SIN(L8)*.397821+SIN(L7)*.917463
1530 LET   D1=ATN(D1  /(SQR(1-D1↑2)))
1531 G1=50+.5+((D1)/(.792))*D5
1532 G2=80+((D1)/(.808))*D5
1533 G3=141.5-((D1)*(.738)*D5)
1534 G4=170.5-((D1)*(.857)*D5)
1540 A2=COS(L7)*COS(L8)/COS(D1)
1550  A1=(COS(L7)*SIN(L8)*.917463-SIN(L7)*.397821)/COS(D1)
1560 A=ATN(A1/A2)
1570 GOSUB 1870
1580  R1=A
1590  L1=.065709822*T1
```

```
1600 L=T*24*1.002738+6.646055+(L1-INT(L1/24)*24)
1610  L=(L-INT(L/24)*24)
1630  G=(L/24)*P5-R1
1640 IF G<P5 THEN 1670
1650 G=G-P5
1660 GOTO 1710
1670 IF G<0 THEN 1690
1680 GOTO 1710
1690 G=G+P5
1710  H=L6-G
1730  E3=COS(L5)*COS(H)*COS(D1)+SIN(D1)*SIN(L5)
1740  E2=SQR(1-(E3*E3))
1750 E=ATN((E3/E2)-(1/(61.33*E2)))
1755 F=ATN(E3/E2)
1760 IF E<0 THEN 2178
1770 IF E>I6*R5 THEN 2178
1790 A2=SIN(D1)/(COS(L5)*COS(F))
1800 A2=A2-(SIN(L5)/COS(L5))*(SIN(F)/COS(F))
1810  A1=SIN(L5)*SIN(D1)+COS(L5)*COS(D1)*COS(H)
1820 A1=(SIN(H)*COS(D1))/SQR(1-A1↑2)
1830  A=ATN(A1/A2)
1840 GOSUB 1870
1850 GOTO 2020
1870 IF A=0 THEN 1890
1880 GOTO 1930
1890 IF A2<0 THEN 1910
1900 GOTO 2010
1910  A=P5/2
1920 GOTO 2010
1930 IF A>0 THEN 1990
1940 IF A2<0 THEN 1970
1950  A=P5+A
1960 GOTO 2010
1970  A=P5+(A-P5/2)
1980 GOTO 2010
1990 IF A2=>0 THEN 2010
2000  A=A+P5/2
2010 RETURN
2020 IF (T-I1)>(2*I)/1440 THEN 2040
2030 GOTO 2050
2040 PRINT
2050 IF INT(B+.5)>9 THEN 2080
2060 S$="   "
2070 GOTO 2142
2080 IF INT(B+.5)>99 THEN 2110
2090 S$="  "
2100 GOTO 2142
2110 IF INT(B+.5)>999 THEN 2140
2120 S$=" "
2130 GOTO 2142
2140 S$=""
2142 Z1=FNA(A)
2144 Z2=FNA(E)
2146 Z3=FNA(G)
2148 Z4=FNA(D1)
2150 IF Z4<0 THEN 2163
2151 IF Z3<G1 THEN 2163
2152 IF Z3>G2 THEN 2154
2153 GOTO 2157
2154 IF Z3<G3 THEN 2159
```

```
2155 IF Z3 >G4 THEN 2163
2156 GOTO 2161
2157 Y$="U"
2158 GOTO 2170
2159 Y$="W"
2160 GOTO 2170
2161 Y$="J"
2162 GOTO 2170
2163 Y$=" "
2170 PRINT S$;STR$(INT(B+.5)),Z1,Z2,Z3,Z4;Y$
2176 I1=T
2178 B=B+I
2180  Z=(B-INT(B/100)*100)-60
2190 IF Z<0 THEN 1290
2200  B=INT(B/100)*100+100+Z
2210 GOTO 1290
2220 NEXT N
2230  N=0
2240 PRINT
2260 PRINT
2270 PRINT"DO YOU WANT MORE INFORMATION";
2280 INPUT D$
2290 IF D$="YES" THEN 560
2300 END
OK

-
```

The following two scientific calculator programs were contributed by Shelby Ennis, W4WNH/8 and William Dayton, WA8BAH.

The two equations used for these programs are:

$$\text{Elevation} = \sin^{-1}\left((\cos(\text{GHA-LONG}) \cdot \text{cosLAT} \cdot \text{cosDEC}) + (\text{sinLAT} \cdot \text{sinDEC})\right)$$

$$\text{Azimuth} = \cos^{-1}\left(\left(\frac{\text{sinDEC}}{\text{cosEL} \cdot \text{cosLAT}}\right) - (\text{tanLAT} \cdot \text{tanEL})\right)$$

HP-55 Program Form

Title: EME Antenna AZ and EL from Moon's GHA and Declination Page 1 of 2

Press [BST] in RUN mode switch to PRGM mode. Then key in the program.

DISPLAY LINE	DISPLAY CODE	KEY ENTRY	X	Y	Z	T	COMMENTS	REGISTERS
00								R0 OUTPUT Elevation Degrees
01	34	RCL						
02	08	8						R1 NOT USED
03	71	X						
04	34	RCL						R2 360 constant
05	05	5						
06	71	X						R3 Input Long. Degrees
07	34	RCL						
08	07	7						R4 sin Lat. Degrees
09	34	RCL						
10	04	4						R5 cos Lat. Degrees
11	71	X						
12	61	+						R6 tan Lat Degrees
13	32	G						
14	12	sin						R7 sin Decl. Degrees
15	33	STO						
16	00	0						R8 cos Decl. Degrees
17	31	F						
18	14	tan						R9 sin(GHA -Long.)
19	34	RCL						
20	06	6						R0 NOT USED
21	71	X						
22	34	RCL						R1 NOT USED
23	07	7						
24	34	RCL						R2 NOT USED
25	05	5						
26	81	↓						R3 NOT USED
27	34	RCL						
28	00	0						R4 NOT USED
29	31	F						
30	13	cos						R5 NOT USED
31	81	↓						
32	22	X⇆Y						R6 NOT USED
33	51	-						
34	32	G						R7 NOT USED
35	13	cos						
36	00	0						R8 NOT USED
37	34	RCL						
38	09	9						R9 NOT USED
39	31	F						
40	-48	X≤Y-48						
41	23	R↓						
42	23	R↓						
43	34	RCL						
44	02	2						
45	22	X⇆Y						
46	51	-						
47	-00	GTO-00						
48	23	R↓						
49	23	R↓						

HP-55 User Instructions

Title: EME Antenna AZ and EL from Moon's GHA and Declination Page 2 of 2

Programmer: Shelby Ennis W4WNH/8 and William Dayton WA8BAH

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS				OUTPUT DATA/UNITS
1	Enter program						
2	Store constant and	360	STO	2			360
	antenna location	Long. Deg.	STO	3			Long.
	data	Lat. Deg.	Enter↓	Enter↑	F	sin	sin Lat.
			STO	4	R↓	F	Lat.
			cos	STO	5	R↓	Lat.
			F	tan	STO	6	tan Lat.
3a	Store Moon data	Declination (Deg.Min.Sec.)		F	H←		Degrees
3b	[enter South Dec. as negative(use "CHS")]	Dec. Deg.	Enter↑	F	sin	STO	sin Dec.
			7	R↓	F	cos	cos Dec.
			STO	8			
3c		GHA (Deg.Min.Sec)	F	H←			Degrees
		GHA Deg.	RCL	3	-	Enter↑	
			F	sin	STO	9	sin(GHA-Long.)
			R↓	F	cos		cos(GHA-Long.)
4	Run program		BST	R/S			Azimuth °
			RCL	0			Elevation°
	For new Moon data, go to step 3						
	to run program again w/same data		RCL	9	G	sin	
			F	cos	BST	R/S	Azimuth°
			RCL	0			Elevation°

Note: Steps 3a and 3c take data in degrees, minutes and seconds and convert to decimal degrees. If the data are already in decimal degrees, enter data at step 3b and 3d directly. The Nautical Almanac gives the Moon data in degrees, minutes and tenths of minutes. Rounding off the minutes causes negligible additional error to the plus or minus one degree error of the program. Example 1: Degree, minute, seconds S26° 53' 42" is entered as (negative because of south) -26.5342°. Example 2: Degrees, minutes and tenths of minutes, S26° 53.7' is entered as -26.54°. The program does not take into account your antenna height, or "beam bending".

HP-25 Program Form

Title EME Antenna AZ and EL from Moon's GHA and Declination Page 1 of 2

Switch to PRGM mode, press [f] [PRGM], then key in the program.

DISPLAY LINE	DISPLAY CODE	KEY ENTRY	X	Y	Z	T	COMMENTS
00							
01	23-02	STO 2					
02	21	X⇄Y					
03	24-04	RCL 4					
04	41	-					
05	31	ENTER↑					
06	14-04	F sin					
07	23-06	STO 6					
08	22	R↓					
09	14-05	F cos					
10	24-02	RCL 2					
11	14-05	F cos					
12	61	X					
13	24-05	RCL 5					
14	14-05	F cos					
15	61	X					
16	24-02	RCL 2					
17	14-04	F sin					
18	24-05	RCL 5					
19	14-04	F sin					
20	61	X					
21	51	+					
22	15-04	Gsin-1					
23	23-07	STO 7					
24	74	R/S					
25	14-06	F tan					
26	24-05	RCL 5					
27	14-06	F tan					
28	61	X					
29	24-02	RCL 2					
30	14-04	F sin					
31	24-05	RCL 5					
32	14-05	F cos					
33	71	÷					
34	24-07	RCL 7					
35	14-05	F cos					
36	71	÷					
37	21	X⇄Y					
38	41	-					
39	15-05	Gcos-1					
40	23-00	STO 0					
41	24-06	RCL 6					
42	15-41	G x<0					
43	13-48	GTO 48					
44	24-03	RCL 3					
45	24-00	RCL 0					
46	41	-					
47	23-00	STO 0					
48	24-00	RCL 0					
49							

REGISTERS	
R0	output AZ
R1	not used
R2	Decl.
R3	360
R4	input Long.
R5	Input Lat.
R6	(prog) sin L.H.A.
R7	output Elev.

HEWLETT hp PACKARD

HP-25 Program Form

Title EME Antenna AZ and EL from Moon's GHA and Declination Page 2 of 2

Programmer

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS				OUTPUT DATA/UNITS
1	Enter program						
2	Store constant	360	STO	3			360
3	Store QTH position	Long.	G	H→	STO	4	Long.°
	Note: Keystrokes "G" &	Lat.	G	H→	STO	5	Lat.°
	"H←" convert Deg.,						
	Min., Sec., to Decimal						
	Degrees. Omit these						
	keystrokes if data are						
	already in Decimal						
	Degrees.						
	Enter Latitude as						
	"negative" if south of						
	the equator. (Use						
	"CHS" key).						
4	Key in Moon Data	GHA	G	H→	ENTER↑		
	(Enter southern Dec.,	Dec.	G	H→			
	as "negative").						
5	Run program		R/S				El.°
			R/S				Az.°
	El. is also stored in						
	Register #7, Az in #0.						
	For another Moon						
	position go to step 4.						